Soluzioni del compito di Calcolo Numerico del 29 luglio 2003

Enrico Bertolazzi

```
> restart:
> with(LinearAlgebra):
```

Esercizio 1

matrice identita e vettori canonici della base usati per i conti successivi

```
> e1,e2,e3,e4 := <1,0,0,0>,<0,1,0,0>,<0,0,1,0>,<0,0,0,1>;
```

$$e1, e2, e3, e4 := \begin{bmatrix}1\\0\\0\\0\end{bmatrix}, \begin{bmatrix}0\\1\\0\\0\end{bmatrix}, \begin{bmatrix}0\\0\\1\\0\end{bmatrix}, \begin{bmatrix}0\\0\\0\\1\end{bmatrix}$$

Matrice identità

```
> ID := <e1|e2|e3|e4>;
```

$$ID := \begin{bmatrix}1 & 0 & 0 & 0\\0 & 1 & 0 & 0\\0 & 0 & 1 & 0\\0 & 0 & 0 & 1\end{bmatrix}$$

Matrice iniziale

```
> A := Matrix([[-3, 2, 11, 48],
               [12, -24, -36, -48],
               [-6, 24, -6, -12],
               [-4, 2, 36, 10]]);
```

$$A := \begin{bmatrix}-3 & 2 & 11 & 48\\12 & -24 & -36 & -48\\-6 & 24 & -6 & -12\\-4 & 2 & 36 & 10\end{bmatrix}$$

```
> A0 := A :
```

Matrice di permutazione

```
> P1 := <e2|e1|e3|e4>;
```

$$P1 := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Scambio la prima con la seconda riga

```
> P1A0 := P1.A0;
```

$$P1A0 := \begin{bmatrix} 12 & -24 & -36 & -48 \\ -3 & 2 & 11 & 48 \\ -6 & 24 & -6 & -12 \\ -4 & 2 & 36 & 10 \end{bmatrix}$$

Matrice di eliminazione

```
> L1 := ID-<0,P1A0[2,1],P1A0[3,1],P1A0[4,1]>.Transpose(e1)/P1A0[1,1];
```

$$L1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{1}{4} & 1 & 0 & 0 \\ \frac{1}{2} & 0 & 1 & 0 \\ \frac{1}{3} & 0 & 0 & 1 \end{bmatrix}$$

Primo passo del metodo di Gauss

```
> A1 := L1.P1A0;
```

$$A1 := \begin{bmatrix} 12 & -24 & -36 & -48 \\ 0 & -4 & 2 & 36 \\ 0 & 12 & -24 & -36 \\ 0 & -6 & 24 & -6 \end{bmatrix}$$

Matrice di scambio seconda e terza riga

```
> P2 := <e1|e3|e2|e4>;
```

$$P2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Scambio la seconda con la terza riga

```
> P2A1 := P2.A1;
```

$$P2A1 := \begin{bmatrix} 12 & -24 & -36 & -48 \\ 0 & 12 & -24 & -36 \\ 0 & -4 & 2 & 36 \\ 0 & -6 & 24 & -6 \end{bmatrix}$$

Matrice di eliminazione

```
> L2 := ID-<0,0,P2A1[3,2],P2A1[4,2]>.Transpose(e2)/P2A1[2,2];
```

$$L2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & \frac{1}{3} & 1 & 0 \\ 0 & \frac{1}{2} & 0 & 1 \end{bmatrix}$$

Secondo Passo del metodo di Gauss

```
> A2 := L2.P2A1;
```

$$A2 := \begin{bmatrix} 12 & -24 & -36 & -48 \\ 0 & 12 & -24 & -36 \\ 0 & 0 & -6 & 24 \\ 0 & 0 & 12 & -24 \end{bmatrix}$$

Matrice di scambio terza e quarta riga

```
> P3 := <e1|e2|e4|e3>;
```

$$P3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

Scambia terza e quarta riga

```
> P3A2 := P3.A2;
```

$$P3A2 := \begin{bmatrix} 12 & -24 & -36 & -48 \\ 0 & 12 & -24 & -36 \\ 0 & 0 & 12 & -24 \\ 0 & 0 & -6 & 24 \end{bmatrix}$$

Matrice di eliminazione

```
> L3 := ID-<0,0,0,P3A2[4,3]>.Transpose(e3)/P3A2[3,3];
```

$$L3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & \frac{1}{2} & 1 \end{bmatrix}$$

Terzo passo del metodo di Gauss

```
> A3 := L3.P3A2;
```

$$A3 := \begin{bmatrix} 12 & -24 & -36 & -48 \\ 0 & 12 & -24 & -36 \\ 0 & 0 & 12 & -24 \\ 0 & 0 & 0 & 12 \end{bmatrix}$$

Matrici P, L, U

```
> P := P3.P2.P1:
  L := P.(L3.P3.L2.P2.L1.P1)^(-1):
  U := A3:
  P, L, U ;
```

$$\begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{-1}{2} & 1 & 0 & 0 \\ \frac{-1}{3} & \frac{-1}{2} & 1 & 0 \\ \frac{-1}{4} & \frac{-1}{3} & \frac{-1}{2} & 1 \end{bmatrix}, \begin{bmatrix} 12 & -24 & -36 & -48 \\ 0 & 12 & -24 & -36 \\ 0 & 0 & 12 & -24 \\ 0 & 0 & 0 & 12 \end{bmatrix}$$

Controllo

```
> R := P.A - L.U;
```

$$R := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Calcolo la soluzione del primo problema

```
> b  := < 10, -48, 12, 34>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  b, Pb, z, x ;
```

$$\begin{bmatrix} 10 \\ -48 \\ 12 \\ 34 \end{bmatrix}, \begin{bmatrix} -48 \\ 12 \\ 34 \\ 10 \end{bmatrix}, \begin{bmatrix} -48 \\ -12 \\ 12 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 1 \\ 0 \end{bmatrix}$$

Calcolo la soluzione del secondo problema

```
> b  := < 168, -240, -24, 104>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  b, Pb, z, x ;
```

$$\begin{bmatrix} 168 \\ -240 \\ -24 \\ 104 \end{bmatrix}, \begin{bmatrix} -240 \\ -24 \\ 104 \\ 168 \end{bmatrix}, \begin{bmatrix} -240 \\ -144 \\ -48 \\ 36 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 2 \\ 3 \end{bmatrix}$$

Esercizio 2

libero la variabile x dell'esercizio precedente

```
> x := 'x';
```

$$x := x$$

Funzione di partenza

```
> f := x -> x - 1/(1+x^2);
```

$$f := x \to x - \frac{1}{1 + x^2}$$

Derivata prima

```
> df := D(f);
```

$$df := x \to 1 + \frac{2\,x}{(1 + x^2)^2}$$

Metodo di Newton

```
> Newton := unapply( factor(x - f(x) / df(x)), x);
```

$$Newton := x \to \frac{3\,x^2 + 1}{1 + 2\,x^2 + x^4 + 2\,x}$$

Metodo delle secanti

```
> Secanti := unapply( factor( (y*f(x)-x*f(y)) / (f(x)-f(y))), x, y);
```

$$Secanti := (x, y) \to \frac{x^2 + y\,x + 1 + y^2}{x^2 + x^2\,y^2 + x + 1 + y^2 + y}$$

Metodo di Newton a partire da x0 = 1:

```
> x0 := 1:
  x1 := Newton(x0):
  x2 := Newton(x1):
  x3 := Newton(x2):
  x4 := Newton(x3):
  <x0,x1,x2,x3,x4>,
  evalf(<x0,x1,x2,x3,x4>) ;
```

$$\begin{bmatrix} 1 \\ \frac{2}{3} \\ \frac{189}{277} \\ \frac{7054924634}{10339494587} \\ \frac{27391347543685458748085029693768512086653}{40143970962363781787578928260040051883429} \end{bmatrix}, \begin{bmatrix} 1. \\ 0.6666666667 \\ 0.6823104693 \\ 0.6823278038 \\ 0.6823278038 \end{bmatrix}$$

Metodo delle secanti a partire da x0 = 1 x1 = 2 ;

```
> x0 := 1:
  x1 := 2:
  x2 := Secanti(x0,x1):
  x3 := Secanti(x1,x2):
  x4 := Secanti(x2,x3):
  <x0,x1,x2,x3,x4>,
  evalf(<x0,x1,x2,x3,x4>) ;
```

$$\left[\begin{array}{r}1\\2\\\frac{8}{13}\\\frac{1117}{1607}\\\frac{999252034}{1464355061}\end{array}\right], \left[\begin{array}{r}1.\\2.\\0.6153846154\\0.6950840075\\0.6823837064\end{array}\right]$$

Esercizio 3

Punti di interpolazione:

```
> X := [0,1,2,-1,-2];
  Y := [1,3,7,1,3];
```

$$X := [0, 1, 2, -1, -2]$$

$$Y := [1, 3, 7, 1, 3]$$

Polinomio interpolante:

```
> interp( X, Y, 'z');
```

$$z^2 + z + 1$$

Costruzione delle differenze divise di ordine 0

```
> f1  := Y[1];
  f2  := Y[2];
  f3  := Y[3];
  f4  := Y[4];
  f5  := Y[5];
```

$$f1 := 1$$

$$f2 := 3$$

$$f3 := 7$$

$$f4 := 1$$

$$f5 := 3$$

Differenze divise

```
> f12 := (f2-f1)/(X[2]-X[1]);
  f23 := (f3-f2)/(X[3]-X[2]);
  f34 := (f4-f3)/(X[4]-X[3]);
  f45 := (f5-f4)/(X[5]-X[4]);
```

$$f12 := 2$$

$$f23 := 4$$

$$f34 := 2$$

$$f45 := -2$$

Differenze divise seconde

```
> f123 := (f23-f12)/(X[3]-X[1]);
  f234 := (f34-f23)/(X[4]-X[2]);
  f345 := (f45-f34)/(X[5]-X[3]);
```

$$f123 := 1$$

$$f234 := 1$$

$$f345 := 1$$

Differenze divise terze

```
> f1234 := (f234-f123)/(X[4]-X[1]);
  f2345 := (f345-f234)/(X[5]-X[2]);
```

$$f1234 := 0$$

$$f2345 := 0$$

Differenze divise quarte

```
> f12345 := (f2345-f1234)/(X[5]-X[1]);
```

$$f12345 := 0$$

Polinomi della base

```
> w0 := 1 ;
  w1 := x-X[1] ;
  w2 := expand(w1 * ( x - X[2])) ;
  w3 := expand(w2 * ( x - X[3])) ;
  w4 := expand(w3 * ( x - X[4])) ;
```

$$w0 := 1$$

$$w1 := x$$

$$w2 := x^2 - x$$

$$w3 := x^3 - 3\,x^2 + 2\,x$$

$$w4 := x^4 - 2\,x^3 - x^2 + 2\,x$$

Polinomio interpolante

```
> p := f1*w0 + f12 * w1 + f123 * w2 + f1234 * w3 + f12345 * w4 ;
```

$$p := x^2 + x + 1$$

Esercizio 4

Funzione da integrare:

```
> f := x -> (1+x^2)*cos(2*x) ;
```

$$f := x \rightarrow (1 + x^2)\cos(2\,x)$$

Integrale esatto

```
> a,b := -1,2:
  int(f(x),x=a..b);
  esatto := evalf(%);
```

$$\frac{3}{4}\sin(2) + \frac{1}{2}\cos(2) + \frac{9}{4}\sin(4) + \cos(4)$$

$$esatto := -1.882549583$$

Derivata seconda della funzione

```
> ddf := (D@D)(f) ;
  collect(ddf(x),cos(2*x)) ;
```

$$ddf := x \to 2\cos(2\,x) - 8\,x\sin(2\,x) - 4\,(1 + x^2)\cos(2\,x)$$

$$(-2 - 4\,x^2)\cos(2\,x) - 8\,x\sin(2\,x)$$

Possibile costante che maggiore il modulo della derivata seconda

```
> M := 2 + 16 + 4*(1+4) ;
```

$$M := 38$$

Verifica grafica:

```
> plot(abs(ddf(x)),x=a..b) ;
```



Calcolo in numero minimo stimato di intervalli

```
> err := 10^(-5) = (b-a)^3/(12*n^2)*M ;
```

$$err := \frac{1}{100000} = \frac{171}{2\,n^2}$$

```
> isolate(err,n^2) ;
```

$$n^2 = 8550000$$

```
> evalf(sqrt(rhs(%)));
```

$$2924.038304$$

Intervalli stimati: 2925

```
> h := (b-a)/6 ;
```

$$h := \frac{1}{2}$$

```
> trap := evalf(h*(f(a)+f(b))/2+h*sum(f(a+h*i),i=1..5)) ;
```

$$trap := -1.874634706$$

```
> trap - esatto ;
```

$$0.007914877$$

Esercizio 5

```
> h := 'h';
```

$$h := h$$

Funzione da integrare

```
> f := (x,y) -> x^2 - 3*y ;
```

$$f := (x, y) \rightarrow x^2 - 3\, y$$

Derivata prima

```
> dy := f(x,y) ;
```

$$dy := x^2 - 3\, y$$

Derivata seconda

```
> ddy := diff(dy,x)+diff(dy,y)*f(x,y) ;
```

$$ddy := 2\, x - 3\, x^2 + 9\, y$$

Derivata terza

```
> dddy := diff(ddy,x)+diff(ddy,y)*f(x,y) ;
```

$$dddy := 2 - 6\, x + 9\, x^2 - 27\, y$$

Costruzione della serie di Taylor troncata

```
> ynew := unapply( y+dy*h+ddy*h^2/2+dddy*h^3/6, x, y, h) ;
```

$$ynew := (x, y, h) \rightarrow y + (x^2 - 3\, y)\, h + \frac{1}{2}(2\, x - 3\, x^2 + 9\, y)\, h^2 + \frac{1}{6}(2 - 6\, x + 9\, x^2 - 27\, y)\, h^3$$

Calcolo alcuni passi

```
> x0, y0 := 1, 2;
```

$$x0, y0 := 1, 2$$

```
> h := 0.2 ;
```

$$h := 0.2$$

```
> x1 := x0 + h ;
  y1 := ynew(x0, y0, h) ;
```

$$x1 := 1.2$$

$$y1 := 1.274666667$$

```
> x2 := x1 + h ;
  y2 := ynew(x1, y1, h) ;
```

$$x2 := 1.4$$

$$y2 := 0.9533653334$$

```
> x3 := x2 + h ;
  y3 := ynew(x2, y2, h) ;
```

$$x3 := 1.6$$

$$y3 := 0.8640174081$$

Esercizio 6

Definizione del problema

```
> p := x -> x-1 ;
  q := x -> -1 ;
  r := x -> x^2 - 2 * x ;
  xa, ya := -1, 1 ;
  xb, yb := 3, 13 ;
```

$$p := x \to x - 1$$

$$q := x \to -1$$

$$r := x \to x^2 - 2\,x$$

$$xa, ya := -1, 1$$

$$xb, yb := 3, 13$$

Differenze finite:

```
> n := 4 ;
  h := (xb-xa) / n ;
  x[0] := xa ;
  x[1] := xa + h ;
  x[2] := xa + 2*h ;
  x[3] := xa + 3*h ;
  x[4] := xa + 4*h ;
```

$$n := 4$$

$$h := 1$$

$$x_0 := -1$$

$$x_1 := 0$$

$$x_2 := 1$$

$$x_3 := 2$$

$$x_4 := 3$$

```
> eq := k -> (y[k+1]-2*y[k]+y[k-1])/h^2 +
             p(x[k]) * (y[k+1]-y[k-1])/(2*h) +
             q(x[k]) * y[k] - r(x[k]) ;
```

$$eq := k \rightarrow \frac{y_{1+k} - 2\,y_k + y_{k-1}}{h^2} + \frac{p(x_k)\,(y_{1+k} - y_{k-1})}{2\,h} + q(x_k)\,y_k - r(x_k)$$

Equazioni risultanti

```
> eq1 := eq(1) ;
  eq2 := eq(2) ;
  eq3 := eq(3) ;
```

$$eq1 := \frac{1}{2} y_2 - 3\,y_1 + \frac{3}{2} y_0$$

$$eq2 := y_3 - 3\,y_2 + y_1 + 1$$

$$eq3 := \frac{3}{2} y_4 - 3\,y_3 + \frac{1}{2} y_2$$

Estraggo il sistema lineare dalle equazioni

```
> A := linalg[genmatrix]([eq1,eq2,eq3],[y[1],y[2],y[3]],'b'):
  A := convert(A,Matrix);
```

$$A := \begin{bmatrix} -3 & \frac{1}{2} & 0 \\ 1 & -3 & 1 \\ 0 & \frac{1}{2} & -3 \end{bmatrix}$$

```
> b := Transpose(convert(b,Vector));
```

$$b := \begin{bmatrix} -\frac{3}{2} y_0 \\ -1 \\ -\frac{3}{2} y_4 \end{bmatrix}$$

Sostituisco le condizioni al contorno

```
> b := subs(y[0]=ya, y[4]=yb, b) ;
```

$$b := \begin{bmatrix} \frac{-3}{2} \\ -1 \\ \frac{-39}{2} \end{bmatrix}$$

Risolvo il sistema lineare

```
> LinearSolve(A,b) ;
```

$$\begin{bmatrix} 1 \\ 3 \\ 7 \end{bmatrix}$$

Esercizio 7

Funzione da integrare

```
> f := x -> 1 + x^2;
```

$$f := x \to 1 + x^2$$

Estremi di integrazione

```
> a,b := -1,3 ;
```

$$a, b := -1, 3$$

Integrale esatto

```
> int(f(x),x=a..b);
  evalf(%) ;
```

$$\frac{40}{3}$$

$$13.33333333$$

Nodi e pesi

```
> X := [-0.7746, 0, 0.7746 ];
  W := [0.5555, 0.8888, 0.5555 ] ;
```

$$X := [-.7746, 0, 0.7746]$$

$$W := [0.5555, 0.8888, 0.5555]$$

```
> c := 1 ;
```

$$c := 1$$

Integrale approssimato nel primo intervallo [a,c]

Trasformazione dei nodi

```
> XT := [seq((a+c)/2+X[i]*(c-a)/2,i=1..3)];
```

$$XT := [-.7746000000, 0, 0.7746000000]$$

```
> int1 := ((c-a)/2)*sum(W[i]*f(XT[i]),i=1..3);
```

$$int1 := 2.666405732$$

Integrale approssimato nel primo intervallo [c,b]

Trasformazione dei nodi

```
> XT := [seq((c+b)/2+X[i]*(b-c)/2,i=1..3)];
```

$$XT := [1.225400000, 2, 2.774600000]$$

```
> int2 := ((b-c)/2)*sum(W[i]*f(XT[i]),i=1..3) ;
```

$$int2 := 10.66560573$$

Risultato finale

```
> int1+int2 ;
```

$$13.33201146$$

Esercizio 8

Matrice del sistema

```
> L  := <<0,0,-1>|<0,0,-1>|<0,0,0>>;
  DG := <<2,0,0>|<0,4,0>|<0,0,2>>;
  U  := <<0,0,0>|<-1,0,0>|<-1,-1,0>>;
  A  := DG+L+U ;
```

$$L := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ -1 & -1 & 0 \end{bmatrix}$$

$$DG := \begin{bmatrix} 2 & 0 & 0 \\ 0 & 4 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

$$U := \begin{bmatrix} 0 & -1 & -1 \\ 0 & 0 & -1 \\ 0 & 0 & 0 \end{bmatrix}$$

$$A := \begin{bmatrix} 2 & -1 & -1 \\ 0 & 4 & -1 \\ -1 & -1 & 2 \end{bmatrix}$$

Termine noto

```
> b := <1,-1,1>;
```

$$b := \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix}$$

libera x, y, z

```
> x := 'x' ;
  y := 'y' ;
  z := 'z' ;
```

$$x := x$$

$$y := y$$

$$z := z$$

Metodo Iterativo

```
> iter := (DG+L).<x[k+1],y[k+1],z[k+1]> + U.<x[k],y[k],z[k]> - b ;
```

$$iter := \begin{bmatrix} 2\,x_{k+1} - y_k - z_k - 1 \\ 4\,y_{k+1} - z_k + 1 \\ -x_{k+1} - y_{k+1} + 2\,z_{k+1} - 1 \end{bmatrix}$$

```
> isolate(iter[1]=0,x[k+1]);
  isolate(iter[2]=0,y[k+1]);
  isolate(iter[3]=0,z[k+1]);
```

$$x_{k+1} = \frac{1}{2} y_k + \frac{1}{2} z_k + \frac{1}{2}$$

$$y_{k+1} = \frac{1}{4} z_k - \frac{1}{4}$$

$$z_{k+1} = \frac{1}{2} x_{k+1} + \frac{1}{2} y_{k+1} + \frac{1}{2}$$

Matrice di iterazione -(D+L)^(-1)U

```
> MI := -(DG+L)^(-1).U;
```

$$MI := \begin{bmatrix} 0 & \frac{1}{2} & \frac{1}{2} \\ 0 & 0 & \frac{1}{4} \\ 0 & \frac{1}{4} & \frac{3}{8} \end{bmatrix}$$

Polinomio caratteristico della matrice di iterazione

```
> chpoly := CharacteristicPolynomial(MI,z);
```

$$chpoly := z^3 - \frac{3}{8} z^2 - \frac{1}{16} z$$

```
> res := solve(chpoly,z);
```

$$res := 0, \frac{1}{2}, \frac{-1}{8}$$

Calcolo il raggio spettrale

```
> seq(abs(res[i]),i=1..3);
```

$$0, \frac{1}{2}, \frac{1}{8}$$

Raggio spettrale = 1/2 < 1 ==> il metodo converge.

Faccio alcune iterate:

```
> P0 := <2,1,2> ;
```

$$P0 := \begin{bmatrix} 2 \\ 1 \\ 2 \end{bmatrix}$$

```
> P1 := (DG+L)^(-1).(b-U.P0):
  P1, evalf(P1) ;
```

$$\begin{bmatrix} 2 \\ \frac{1}{4} \\ \frac{13}{8} \end{bmatrix}, \begin{bmatrix} 2. \\ 0.2500000000 \\ 1.625000000 \end{bmatrix}$$

```
> P2 := (DG+L)^(-1).(b-U.P1) :
  P2, evalf(P2) ;
```

$$\begin{bmatrix} \frac{23}{16} \\ \frac{5}{32} \\ \frac{83}{64} \end{bmatrix}, \begin{bmatrix} 1.437500000 \\ 0.1562500000 \\ 1.296875000 \end{bmatrix}$$

Calcolo i residui

```
> R1 := b - A . P1 :
  R2 := b - A . P2 :
  R1, evalf(R1) ;
  R2, evalf(R2) ;
```

$$\begin{bmatrix} \frac{-9}{8} \\ \frac{-3}{8} \\ 0 \end{bmatrix}, \begin{bmatrix} -1.125000000 \\ -.3750000000 \\ 0. \end{bmatrix}$$

$$\begin{bmatrix} \frac{-27}{64} \\ \frac{-21}{64} \\ 0 \end{bmatrix}, \begin{bmatrix} -.4218750000 \\ -.3281250000 \\ 0. \end{bmatrix}$$

Norma infinito

```
> max(seq(abs(R1[i]),i=1..3));
  max(seq(abs(R2[i]),i=1..3));
  evalf(%%),evalf(%) ;
```

$$\frac{9}{8}$$

$$\frac{27}{64}$$

$$1.125000000, 0.4218750000$$

Norma 1

```
> add(abs(R1[i]),i=1..3);
  add(abs(R2[i]),i=1..3);
  evalf(%%),evalf(%) ;
```

$$\frac{3}{2}$$

$$\frac{3}{4}$$

$$1.500000000, 0.7500000000$$

Norma 2

```
> sqrt(add(abs(R1[i])^2,i=1..3));
  sqrt(add(abs(R2[i])^2,i=1..3));
  evalf(%%),evalf(%) ;
```

$$\frac{3}{8}\sqrt{10}$$

$$\frac{3}{64}\sqrt{130}$$

$$1.185854122, 0.5344572305$$

Esercizio 9

Tabella dei punti

```
> X := [-2,-2,-1,0,2,1,3,-3,2,5];
  Y := [-1,-1,0,1,3,2,4,-2,3,6];
```

$$X := [-2, -2, -1, 0, 2, 1, 3, -3, 2, 5]$$

$$Y := [-1, -1, 0, 1, 3, 2, 4, -2, 3, 6]$$

```
> n := nops(X) ;
```

$$n := 10$$

```
> SX   := add(X[i],i=1..n) ;
  SX2  := add(X[i]^2,i=1..n) ;
  SX3  := add(X[i]^3,i=1..n) ;
  SX4  := add(X[i]^4,i=1..n) ;
  SY   := add(Y[i],i=1..n) ;
  SXY  := add(X[i]*Y[i],i=1..n) ;
```

```
SX2Y := add(X[i]^2*Y[i],i=1..n) ;
```

$$SX := 5$$

$$SX2 := 61$$

$$SX3 := 125$$

$$SX4 := 853$$

$$SY := 15$$

$$SXY := 66$$

$$SX2Y := 186$$

```
> A := <<n,SX,SX2>|<SX,SX2,SX3>|<SX2,SX3,SX4>>;
```

$$A := \begin{bmatrix} 10 & 5 & 61 \\ 5 & 61 & 125 \\ 61 & 125 & 853 \end{bmatrix}$$

```
> b := <SY,SXY,SX2Y> ;
```

$$b := \begin{bmatrix} 15 \\ 66 \\ 186 \end{bmatrix}$$

```
> res := A^(-1).b ;
```

$$res := \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix}$$

Soluzione

```
> x := 'x' :
  p := 'p' :
```

```
> p := res[1]+ x * res[2] + x^2 * res[3] ;
```

$$p := 1 + x$$